

```

1 function SynapseAnalysis(input, output, filename) {
2   open(input + filename);
3   run("Z Project...", "projection=[Max Intensity]");
4   run("Make Composite");
5   saveAs("Tiff", output + "1_MAXINT_COMP " + filename);
6   imageTitle=getTitle();
7   setBatchMode(false);
8   waitForUser("Select Region of Interest before clicking OK");
9   run("ROI Manager...");
10  roiManager("Add");
11  setBatchMode(true);
12  run("Split Channels");
13  selectWindow("C4-" + imageTitle);
14  close();
15  selectWindow("C3-" + imageTitle);
16  close();
17  selectWindow("C1-" + imageTitle);
18  close();
19  selectWindow("C2-" + imageTitle);
20  saveAs("Tiff", output + "2_GREEN " + filename);
21  roiManager("Select", 0);
22  run("Clear Outside");
23  saveAs("Tiff", output + "3_CROPPED_ROI " + filename);
24  run("Despeckle");
25  run("Subtract Background...", "rolling=10");
26  setAutoThreshold("MaxEntropy dark");
27  //run("Threshold...");
28  call("ij.plugin.frame.ThresholdAdjuster.setMode", "B&W");
29  run("Convert to Mask");
30  run("Convert to Mask");
31  run("Watershed");
32  saveAs("Tiff", output + "4_IMG_CORRECTION " + filename);
33  run("Analyze Particles...", "size=10-200 pixel show=Outlines display clear include summarize add");
34  saveAs("Results", output + filename + ".csv");
35  roiManager("Save", output + filename + " ROI.zip");
36  roiManager("Delete");
37  saveAs("Tiff", output + "5_FINAL " + filename);
38  close();
39 }

```

We did a max intensity projection of the multiple planes captured in our Z-stacks

A composite of all channels was made to facilitate drawing our Region of Interest. If this is not needed, delete text from Line 4.

Use selection tool to draw a border around your Region of Interest before clicking the OK button.

In our study, we were interested in the green channel; however, you can alter the script to keep whichever channel you are interested in by changing the magenta text in lines 13, 15, 17 and 19.

Region outside the user-defined ROI is cropped out of the image before analysis

The Despeckle function (line 24) reduces noise in an image by replacing each pixel with the median value in its 3 x 3 neighborhood.

The Rolling Ball Background Subtraction (line 25) averages the local background around each pixel. This average is then subtracted from the original image, removing spatial variations of background intensity. The radius around the pixel can be adjusted, but we used the value "10".

A number of thresholding methods are available for use in ImageJ. We recommend testing them to determine which one best captures the particles you want to quantify. You should also specify whether you are quantifying an image with a dark or light background. As we were quantifying immunofluorescent images, our background was "dark", as specified here.

FIJI will convert the image to Black and White for analysis

The Watershed function (line 31) will draw a line between two closely adjacent particles that it interprets as two separate particles, in order to prevent counting them as one particle.

The Particle Analysis (line 33) can detect objects from 0-infinity pixels-squared. For our purposes, we found that limiting the analysis to particles that were 10-200 pixels-squared captured the synapses to quantify.

Results of the particle analysis are saved as a .csv file.

The user-defined ROI is saved as a .zip file, which can be imported into FIJI and overlayed on an image if desired.

The user must specify the folder that contains the images to be analyzed (the input folder) as well as the folder to save the new images into (the output folder). Pathnames should only include forward slashes (/), and backslashes (\) should be changed to forward slashes. Similarly, be sure that all quotation marks throughout the macro are straight (") and not curved. Enclose pathnames in quotation marks.

```

41 input = "enter input directory"
42 output = "enter output directory"
43
44 setBatchMode(true);
45 list = getFileList(input);
46 for (i = 0; i < list.length; i++)
47   SynapseAnalysis(input, output, list[i]);
48 setBatchMode(false);

```

## To use this Macro:

1. Open ImageJ
2. Open the ScriptEditor ("File -> new -> script" OR keyboard shortcut "[ " OR "ctrl + shift + n")
3. Open the SynapseAnalysis.ijm file
4. Modify the script as needed (described above)
5. Hit "Run" button